

Building Custom Linux Systems with the Yocto Project

Robert Joslyn

Schweitzer Engineering Laboratories

Seattle GNU/Linux Conference, 2019

Yocto Project

- It's not an embedded Linux distribution, it creates one for you
- Automates downloading sources, cross-compiling, and assembling images
- Uses tools and metadata co-developed with OpenEmbedded
- Includes a reference distribution called Poky

Yocto Project

- Robust framework for customization
- Auditing
 - Build entire toolchain and final image from source
 - Licensing
 - CVE analysis
- Easier hardware porting
- Community

First Steps

```
$ git clone -b zeus git://git.yoctoproject.org/poky  
$ cd poky  
$ . oe-init-build-env  
$ bitbake core-image-minimal  
$ runqemu
```

Shared State Cache

Downloading and compiling an entire Linux system can take a long time. Bitbake caches build artifacts to help ease the pain.

- Source code is saved to a common directory: `DL_DIR`
- Build artifacts are saved to sstate cache directory: `SSTATE_DIR`
- Can be shared between developers
 - NFS
 - HTTP/HTTPS

Customize

Simple changes can be done in local.conf

local.conf

```
IMAGE_INSTALL_append = " vim"
```

Run the build

```
$ bitbake core-image-minimal  
$ runqemu
```

Customize More

local.conf

```
IMAGE_INSTALL_append = " vim htop"
```

Run the build

```
$ bitbake core-image-minimal
```

Customize More

local.conf

```
IMAGE_INSTALL_append = " vim htop"
```

Run the build


```
$ bitbake core-image-minimal
```

:-(
ERROR: Nothing RPROVIDES 'htop' (but /home/robert/yocto/seagl2019/poky/meta/recipes-core/images/core-image-minimal.bb RDEPENDS on or otherwise requires it)

Bitbake Layers

- Layers are software repositories
- Make reuse easier
- Allow placing recipes into logical groups
- A layer can modify recipes from another layer

OpenEmbedded Layer Index



The screenshot shows a web browser window with the address bar displaying `https://layers.openembedded.org/layerindex/recipe/108561/`. The page title is "OpenEmbedded Layer Index". In the top right corner, there are two buttons: "Submit layer" (blue) and "Log in" (grey). Below the header, there is a breadcrumb trail: `zeus` → `meta-oe` → `htop`. The main content area features the title `htop 2.2.0` in a large font. Below the title is a table with the following data:

Name	htop
Version	2.2.0
Summary	Interactive process viewer
Description	
Section	console/utils
License	GPLv2
Homepage	http://hisham.hm/htop
Recipe file	recipes-support/htop/htop_2.2.0.bb
Layer	meta-oe (zeus branch)

Use the Community

Clone another layer

```
$ git clone -b zeus git://git.openembedded.org/meta-openembedded
```

bblayers.conf

```
BBLAYERS += "/home/robert/yocto/seagl2019/poky/meta-openembedded/meta-oe"
```

Run the build

```
$ bitbake core-image-minimal  
$ runqemu
```

How it Works

- bitbake parses layers for configuration and recipes, then performs the build
- Layers contain recipes
- Recipes build packages
 - Download and patch source code
 - Set configure options
 - Cross-compile
 - rpm, ipk, or deb
- Packages are assembled into images

Recipe Ingredients

```
robert@bucephalus:~/yocto/seagl2019/poky
1 SUMMARY = "Interactive process viewer"
2 HOMEPAGE = "http://hisham.hm/htop"
3 SECTION = "console/utils"
4 LICENSE = "GPLv2"
5 LIC_FILES_CHKSUM = "file://COPYING;md5=c312653532e8e669f30e5ec8bdc23be3"
6
7 DEPENDS = "ncurses"
8
9 SRC_URI = "http://hisham.hm/htop/releases/${PV}/${BP}.tar.gz \
10           file://0001-Use-pkg-config.patch"
11 SRC_URI[md5sum] = "0d816b6beed31edc75babcfbf863ffa8"
12 SRC_URI[sha256sum] = "d9d6826f10ce3887950d709b53ee1d8c1849a70fa38e91d5896ad8cbc6ba3c57"
13
14 inherit autotools pkgconfig
15
16 PACKAGECONFIG ??= "proc \
17                   cgroup \
18                   taskstats \
19                   unicode \
20                   linux-affinity \
21                   delayacct"
22 PACKAGECONFIG[proc] = "--enable-proc,--disable-proc"
23 PACKAGECONFIG[openvz] = "--enable-openvz,--disable-openvz"
24 PACKAGECONFIG[cgroup] = "--enable-cgroup,--disable-cgroup"
25 PACKAGECONFIG[vserver] = "--enable-vserver,--disable-vserver"
26 PACKAGECONFIG[taskstats] = "--enable-taskstats,--disable-taskstats"
27 PACKAGECONFIG[unicode] = "--enable-unicode,--disable-unicode"
28 PACKAGECONFIG[linux-affinity] = "--enable-linux-affinity,--disable-linux-affinity"
29 PACKAGECONFIG[hwloc] = "--enable-hwloc,--disable-hwloc,hwloc"
30 PACKAGECONFIG[setuid] = "--enable-setuid,--disable-setuid"
31 PACKAGECONFIG[delayacct] = "--enable-delayacct,--disable-delayacct,libnl"
32
33 do_configure_prepend () {
34     rm -rf ${S}/config.h
35 }
36
37 "meta-openembedded/meta-oe/recipes-support/htop/htop_2.2.0.bb" 35L, 1367C 1,1 All
```

Modify a Recipe from Another Layer

Don't fork upstream layers. Modify recipes from your own layer.

```
$ bitbake-layers create-layer ../meta-seagl  
$ bitbake-layers add-layer ../meta-seagl
```

Create a .bbappend file.

```
mkdir -p ../meta-seagl/recipes-support/htop  
vi ../meta-seagl/recipes-support/htop/htop_%.bbappend
```

Modify a Recipe from Another Layer

```
meta-seagl/recipes-support/htop/htop_%.bbappend
```

```
PACKAGECONFIG = "proc"
```

```
meta-seagl/recipes-support/htop/htop_%.bbappend
```

```
PACKAGECONFIG_append = " hwloc"
```

```
meta-seagl/recipes-support/htop/htop_%.bbappend
```

```
PACKAGECONFIG_remove = "delayacct"
```

Recipe WORKDIR

```
robert@bucephalus:~/yocto/seagl2019/poky/build _ □ x
robert@bucephalus ~/yocto/seagl2019/poky/build $ ls tmp/work/aarch64-poky-linux/
htop/2.2.0-r0/
0001-Use-pkg-config.patch  pkgdata-sysroot
build                      pseudo
configure.sstate          recipe-sysroot
debugsources.list         recipe-sysroot-native
deploy-rpms               sstate-install-package
htop-2.2.0                sstate-install-packagedata
htop.spec                 sstate-install-package_qa
image                     sstate-install-package_write_rpm
license-destdir           sstate-install-populate_lic
package                   sstate-install-populate_sysroot
packages-split            sysroot-destdir
pkgdata                   temp
pkgdata-pdata-input
robert@bucephalus ~/yocto/seagl2019/poky/build $
```


Package Content

```
robert@bucephalus:~/yocto/seagl2019/poky/build _ □ x
robert@bucephalus ~/yocto/seagl2019/poky/build $ tree tmp/work/aarch64-poky-linux/htop/2.2.0-r0/packages-split/htop
tmp/work/aarch64-poky-linux/htop/2.2.0-r0/packages-split/htop
├─ usr
│   ├── bin
│   │   └─ htop
│   └─ share
│       ├── applications
│       │   └─ htop.desktop
│       ├── pixmaps
│       │   └─ htop.png
5 directories, 3 files
robert@bucephalus ~/yocto/seagl2019/poky/build $
```

Modify Package Content

Remove files from package

```
meta-seagl/recipes-support/htop/htop_%.bbappend
```

```
do_install_append() {  
    rm -r ${D}${datadir}  
}
```

Modify Package Content

```
robert@bucephalus:~/yocto/seagl2019/poky/build _ □ x
robert@bucephalus ~/yocto/seagl2019/poky/build $ tree tmp/work/aarch64-poky-linux/htop/2.2.0-r0/packages-split/htop
tmp/work/aarch64-poky-linux/htop/2.2.0-r0/packages-split/htop
├─ usr
│   └─ bin
│       └─ htop
2 directories, 1 file
robert@bucephalus ~/yocto/seagl2019/poky/build $
```

Larger Changes

- MACHINE

- Select hardware platform
- Impacts compile options, kernel options, and potentially anything that is hardware specific

- DISTRO

- Sets distro-wide policy
- Should applications build GUIs?
- sysvinit or systemd?
- Recipes can inspect DISTRO_FEATURES to enable or disable features

- IMAGE_FEATURES

- Should debug, source, or dev packages be installed?
- Debug or release build?

Booting Real Hardware

local.conf

```
MACHINE = "genericx86-64"  
IMAGE_FSTYPES_append = " wic"  
WKS_FILE = "mkefidisk.wks"
```

More wks examples are in poky/scripts/lib/wic/canned-wks/

Write to disk

```
$ dd if=tmp/deploy/images/genericx86-64/core-image-minimal.wic \  
    of=/dev/sdb bs=1M
```

Changing init

Generally the DISTRO selects the init manager. Poky defaults to sysvinit, but this can be overridden.

local.conf

```
DISTRO_FEATURES_append = " systemd"
VIRTUAL-RUNTIME_init_manager = "systemd"
# Remove initscripts entirely from the image
DISTRO_FEATURES_BACKFILL_CONSIDERED = "sysvinit"
VIRTUAL-RUNTIME_initscripts = ""
```

Tracking Changes

The buildhistory class captures metadata for build and can commit to a git repository. This helps see how changes impact the image as a whole.

- Files in images and packages
- File sizes and permissions
- Package dependencies

local.conf

```
INHERIT += "buildhistory"  
BUILDHISTORY_COMMIT = "1"
```

Resources

Yocto has extensive documentation

- <https://yoctoproject.org>
- <https://yoctoproject.org/docs/latest/mega-manual/mega-manual.html>
- <https://yoctoproject.org/community/mailling-lists/>
- <https://layers.openembedded.org/layerindex/branch/zeus/recipes/>

L^AT_EX source code for this presentation is available on my website:

- <https://git.robertjoslyn.com/seagl2019/>
- robert.joslyn@redrectangle.org